
SAS syntax and output for Chapter 1: Simple Regression
This document assumes some prior familiarity with SAS, particularly how to work with libraries, the data step, and how to import data files. 

Begin by importing the ‘aggression.txt’ data into SAS. The 'data step' command below creates an internal SAS dataset named 'ch1data'. The 'infile' statement specifies the path of the dataset and delimiter used (dlm='09'x indicates that the data file is tab-delimited) while the 'input' statement gives the variable names:
	data ch1data;
 infile 'c:\aggression.dat' dlm='09'x;
 input age BPAQ AISS alcohol BIS NEOc gender NEOo;
run;



The next commands create boxplot and histogram of BPAQ (Figure 1.2).
The 'vbox' statement requests a vertical boxplot, while the 'histogram' statement requests a histogram:
	title "Boxplot of BPAQ";
proc sgplot data=ch1data;
 vbox BPAQ;
run;

	proc sgplot data=ch1data;
 histogram BPAQ;
run; 

The next commands create a more elaborate histogram with super-imposed kernel density smoother (as a solid line) and fitted normal curve (as a dashed line) as shown in Figure 1.3.
The 'density' statement requests a fitted normal curve while the 'type=kernel' option requests the kernel density smoother.
	title "Histogram of BPAQ";
proc sgplot data=ch1data;
 histogram BPAQ;
 density BPAQ/lineattrs=(color=black pattern=2);
 density BPAQ/type=kernel lineattrs=(color=black); 
run; 

The 'means' procedure produces the descriptive statistics in Table 1.1.
The 'maxdec' option specifies the number of decimal places:
	proc means data=ch1data maxdec=2 n mean std median min max range skew kurt;
run;

Simple regression with a single predictor
Scatterplot of BPAQ scores against BIS scores (Figure 1.4).
The variable specified after the 'y' option will go on the y-axis, while the variable specified after the 'x' option goes on the x-axis.
	proc sgplot data=ch1data;
 scatter x=BIS y=BPAQ; 
run;

The 'corr' procedure with 'pearson' and 'fisher' options gives the correlation along with its 95% confidence interval:
	proc corr data=ch1data pearson fisher;
 var BPAQ BIS;
run;

Above, the bias adjusted sample correlation is listed under "correlation estimate". Notice that a Fisher’s z-test is also printed for the null hypothesis that the population correlation equals zero. The bias adjustment is always used when computing p-values under the null-hypothesis in the 'corr' procedure.
We could modify the 'corr' procedure to produce Spearman's rank-order correlation instead:
	proc corr data=ch1data spearman;
 var BPAQ BIS;
run;

The 'scatter' statement in the 'sgplot' procedure can be used to create a scatterplot enhanced with the fitted least-squares regression line as well as a non-parametric LOWESS curve (Figure 1.5).

The 'reg' statement requests the regression line, while the 'lineattrs' option makes the regression line black instead of blue by default. The 'loess' statement requests the LOWESS curve. See what happens if you change the 'smooth' option from '0.2' to '0.8'.
	proc sgplot data=ch1data;
 scatter x=BIS y=BPAQ; 
 reg x=BIS y=BPAQ/lineattrs=(color=black) nomarkers Name="Regression Line";
 loess x=BIS y=BPAQ/lineattrs=(color=blue) smooth=0.2 nomarkers Name="Lowess";
run;

The 'reg' procedure is used to specify linear regression models and estimate them using OLS.
Under the 'model' statement, the variable on the left-hand side of the equal sign (=) operator is the outcome, the variable on the right-hand side is the predictor. :
	proc reg data=ch1data;
 model BPAQ=BIS / clb;
run;
quit;

Above, notice that the estimates match the parameter estimates given in the text, as does the standard error of the estimate for the BIS predictor and the corresponding t-test.
Additionally, the R-squared value matches the coefficient of determination, i.e., the proportion of variance in BPAQ explained by the model.

Simple regression with a dichotomous predictor
The 'means' procedure also includes a 'class' statement which can be used to break down descriptive statistics across a grouping variable:
	proc means data=ch1data maxdec=2 n mean std median min max range skew kurt;
 var BPAQ;
 class gender;
run;

In the output above, it's important to know that results corresponding to gender = 0 pertain to males and results for gender = 1 pertain to females.
Specify, estimate, and get results for the linear model of BPAQ regressed on gender:
	proc reg data=ch1data;
 model BPAQ = gender/ clb;
run;
quit;

Use the 'ttest' procedure to verify that the t-test for the gender regression slope from the regression model matches the between-groups t-test:
	proc ttest data=ch1data;
 class gender;
 var BPAQ;
run;

The value of the t-statistic next to the ‘Pooled’ method (which is based on the homogeneity of variance assumption) and corresponding p-value above match the results for the 'gender' predictor in the regression model (aside from an arbitrary sign flip).
Likewise, the 95% confidence interval is the same (aside from an arbitrary sign flip).

Basic regression diagnostics
Returning to the model regressing BPAQ on BIS, we can plot the residuals of that model, which can be obtained by using the 'r=resid' option. 
So to reproduce Figure 1.8, the code is:
	proc reg data=ch1data;
 model BPAQ = BIS/ clb;
 output out=ch1data r=resid;
run;
quit;

proc sgplot data=ch1data;
 histogram resid;
 density resid/lineattrs=(color=black pattern=2);
 density resid/type=kernel lineattrs=(color=black); 
run;

Plot the residuals against the BIS predictor, enhancing the plot with a LOWESS curve and a spread around the curve (Figure 1.10):
	proc sgplot data=ch1data;
 scatter x=BIS y=resid; 
 loess x=BIS y=resid / lineattrs=(color=blue) smooth=0.3 nomarkers Name="Lowess" CLM;
run;

Option 'h=hatval' in ‘proc reg’ requests hat values, while option 'rstudent= rstudent' requests Studentized residuals and option 'cookd=cookd' requests Cook's distance values. 'id=_n_' creates an id variable from 1 to n. 
Index plots of these can be created simply using the 'sgplot' procedure (Figures 1.13 to 1.15):
	proc reg data=ch1data plots=none;
model BPAQ = BIS;
output out=ch1data h=hatval rstudent=rstudent cookd=cookd;
run;
quit;

data ch1data;
 set ch1data;
 id=_n_;
run;

proc sgplot data=ch1data;
 scatter x=id y=hatval; 
run;

proc sgplot data=ch1data;
 scatter x=id y=rstudent; 
run;

proc sgplot data=ch1data;
 scatter x=id y=cookd; 
run;

